# Maintainer: Balló György <ballogyor+arch at gmail dot com>
# Contributor: Roman Kyrylych <Roman.Kyrylych@gmail.com>
# Contributor: Björn Martensen <bjoern.martensen@gmail.com>
# Contributor: Daniel J Griffiths <ghost1227@archlinux.us>

pkgbase=xcursor-vanilla-dmz
pkgname=('xcursor-vanilla-dmz' 'xcursor-vanilla-dmz-aa')
pkgver=0.4.5
pkgrel=2
arch=('any')
url="http://jimmac.musichall.cz/"
license=('CCPL:by-nc-sa')
makedepends=('xorg-xcursorgen')
source=(http://ftp.de.debian.org/debian/pool/main/d/dmz-cursor-theme/dmz-cursor-theme_$pkgver.tar.xz)
sha256sums=('b719a64ed9c51905743105e5a50b15492695929ab2d6fca2cea760d52ccd1f5c')

build() {
  cd dmz-cursor-theme-$pkgver
  (cd DMZ-White/pngs ; ./make.sh)
  (cd DMZ-Black/pngs ; ./make.sh)
}

package_xcursor-vanilla-dmz() {
  pkgdesc="Vanilla DMZ cursor theme"

  cd dmz-cursor-theme-$pkgver
  install -d "$pkgdir"/usr/share/icons/Vanilla-DMZ/cursors
  cp -a DMZ-White/xcursors/* "$pkgdir"/usr/share/icons/Vanilla-DMZ/cursors
  install -Dm644 DMZ-White/index.theme "$pkgdir"/usr/share/icons/Vanilla-DMZ/index.theme
}

package_xcursor-vanilla-dmz-aa() {
  pkgdesc="Vanilla DMZ AA cursor theme"

  cd dmz-cursor-theme-$pkgver
  install -d "$pkgdir"/usr/share/icons/Vanilla-DMZ-AA/cursors
  cp -a DMZ-Black/xcursors/* "$pkgdir"/usr/share/icons/Vanilla-DMZ-AA/cursors
  install -Dm644 DMZ-Black/index.theme "$pkgdir"/usr/share/icons/Vanilla-DMZ-AA/index.theme
}
